# Problem 3.18

**Problem 3.18.** Use a computer to reproduce Table 3.2 and the associated graphs of entropy, temperature, heat capacity, and magnetization. (The graphs in this section are actually drawn from the analytic formulas derived below, so your numerical graphs won't be quite as smooth.)

Two state paramagnet with 'n' dipoles, nUp spin up, and nDown spin down. The magnetic dipole moment is $\mu$. The external magnetic field has strength B.

Reproduce Table 3.2. Introduced 'scaled' versions of U, M, and S to make table entries dimensionless. Use the N[ ] function to change the huge exact integer expressions into more-readable approximate numerical values.

```
In[33]:= Clear["Global`*"]
```

First, calculate the energy, magnetization, multiplicity, and entropy.

```
In[34]:= Ω[n_, nUp_] := N[n! / (nUp! (n - nUp)!)]

In[35]:= U[n_, nUp_] := μ B (n - 2 nUp); Us[n_, nUp_] := U[n, nUp] / (μ B)

In[36]:= M[n_, nUp_] := N[-U[n, nUp] / B]; Ms[n_, nUp_] := M[n, nUp] / (n μ)

In[37]:= S[n_, nUp_] := k Log[Ω[n, nUp]]; Ss[n_, nup_] := S[n, nup] / k
```

The Temperature and heat capacity calculations involve derivatives. Approximate them by a centered finite-difference scheme for (nUp+1) and (nUp-1). However, that only works if nUp is in the range [1, ntotal-1]. Outside that range, simply call the derivative 0. Also, calculate the scaled versions indicated in the table.

```
In[38]:= Ts[ntotal_, nUp_] :=
          If[0 < nUp < ntotal, (Us[ntotal, nUp + 1] - Us[ntotal, nUp - 1]) / (Ss[ntotal, nUp + 1] - Ss[ntotal, nUp - 1]), 0]

In[39]:= Cvs[ntotal_, nUp_] :=
          If[0 < nUp < ntotal, 1/ntotal * (Us[ntotal, nUp + 1] - Us[ntotal, nUp - 1]) / (Ts[ntotal, nUp + 1] - Ts[ntotal, nUp - 1]), 0]

In[40]:= ntotal = 100;
```

In[41]:=
```
data = Table[{nUp, Us[ntotal, nUp], Ms[ntotal, nUp], Ω[ntotal, nUp],
    Ss[ntotal, nUp], Ts[ntotal, nUp ], Cvs[ntotal, nUp] }, {nUp, ntotal, 0, -1}];
```

Power : Infinite expression $\frac{1}{0.}$ encountered.

Power : Infinite expression $\frac{1}{0.}$ encountered.

Power : Infinite expression $\frac{1}{0.}$ encountered.

General : Further output of Power::infy will be suppressed during this calculation.

In[42]:=
```
(* Show a few sample segments of the full table. *)
n2 = IntegerPart[ntotal / 2];
sample = Flatten[{Range[1, 4], Range[n2 - 1, n2 + 3], Range[-4, -1]}]
```

Out[43]=

{1, 2, 3, 4, 49, 50, 51, 52, 53, -4, -3, -2, -1}

In[44]:=
```
TableForm[data⟦sample⟧,
  TableHeadings → {None, {"nUp", "U/μB", "M/Nμ", "Ω", "S/k", "kT/μB", "CV/Nk"}}]
```

Out[44]//TableForm=

| nUp | U/μB | M/Nμ | Ω | S/k | kT/μB | CV/Nk |
|---|---|---|---|---|---|---|
| 100 | -100 | 1. | 1. | 0. | 0 | 0 |
| 99 | -98 | 0.98 | 100. | 4.60517 | 0.470193 | 0.0738833 |
| 98 | -96 | 0.96 | 4950. | 8.50714 | 0.541394 | 0.309889 |
| 97 | -94 | 0.94 | 161 700. | 11.9935 | 0.599271 | 0.364569 |
| 52 | -4 | 0.04 | $9.32066 \times 10^{28}$ | 66.7046 | 25.2343 | 0.00118777 |
| 51 | -2 | 0.02 | $9.89131 \times 10^{28}$ | 66.764 | 50.4884 | 0 |
| 50 | 0 | 0. | $1.00891 \times 10^{29}$ | 66.7838 | ComplexInfinity | -0.00039613 |
| 49 | 2 | -0.02 | $9.89131 \times 10^{28}$ | 66.764 | -50.4884 | 0 |
| 48 | 4 | -0.04 | $9.32066 \times 10^{28}$ | 66.7046 | -25.2343 | 0.00118777 |
| 3 | 94 | -0.94 | 161 700. | 11.9935 | -0.599271 | 0.364569 |
| 2 | 96 | -0.96 | 4950. | 8.50714 | -0.541394 | 0.309889 |
| 1 | 98 | -0.98 | 100. | 4.60517 | -0.470193 | 0.0738833 |
| 0 | 100 | -1. | 1. | 0. | 0 | 0 |

In[45]:=
```
Splot = ListLinePlot[data〚All, {2, 5}〛, LabelStyle → Larger,
  AxesLabel → {"U/μB", "S/k"}, ImageSize → Scaled[0.8], PlotRange → All]
```

Out[45]=




In[46]:=
```
SetDirectory[NotebookDirectory[]]
Export["Paramagnet-Splot.pdf", Splot]
```

Out[46]=

/Users/dougher a/335/2024/lectures-dev/Ch03

Out[47]=

Paramagnet-Splot.pdf

In[48]:=
```
ListLinePlot[data〚All, {2, 6}〛, LabelStyle → Larger,
  AxesLabel → {"U/μB", "kT/μB"}, ImageSize → Scaled[0.8]]
```

Out[48]=




In[49]:=
```
ListLinePlot[data〚All, {6, 7}〛,
  LabelStyle → Larger, AxesLabel → {"kT/μB", "C/Nk"},
  ImageSize → Scaled[0.8], PlotRange → {{0, 10}, All}]
```

Out[49]=



In[50]:=
```
ListLinePlot[data[[All, {6, 3}]],
 LabelStyle → Larger, AxesLabel → {"kT/μB", "M/Nμ"},
 ImageSize → Scaled[0.8], PlotRange → {{-10, 10}, All}]
```

Out[50]=




In[51]:=
```
ListLinePlot[Map[{1 / #[[6]], #[[3]]} &, data], LabelStyle → Larger,
 AxesLabel → {"μB/kT", "M/Nμ"}, ImageSize → Scaled[0.8], PlotRange → { All}]
```

Power : Infinite expression $\frac{1}{0}$ encountered.

Power : Infinite expression $\frac{1}{0}$ encountered.

ListLinePlot : Value of option PlotRange −> {All} is not All, Full, Automatic, a positive machine number, or an appropriate list of range specifications .

Out[51]=